Exámenes de "Programación funcional con Haskell"

Vol. 11 (Curso 2019-20)

José A. Alonso Jiménez

Grupo de Lógica Computacional Dpto. de Ciencias de la Computación e Inteligencia Artificial Universidad de Sevilla

Sevilla, 25 de noviembre de 2019

Esta obra está bajo una licencia Reconocimiento-NoComercial-CompartirIgual 2.5 Spain de Creative Commons.

Se permite:

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

Bajo las condiciones siguientes:



Reconocimiento. Debe reconocer los créditos de la obra de la manera especificada por el autor.



No comercial. No puede utilizar esta obra para fines comerciales.



Compartir bajo la misma licencia. Si altera o transforma esta obra, o genera una obra derivada, sólo puede distribuir la obra generada bajo una licencia idéntica a ésta.

- Al reutilizar o distribuir la obra, tiene que dejar bien claro los términos de la licencia de esta obra.
- Alguna de estas condiciones puede no aplicarse si se obtiene el permiso del titular de los derechos de autor.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite http://creativecommons.org/licenses/by-nc-sa/2. 5/es/ o envie una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

Índice general

Introducción		5	
1	Exámenes del grupo 1 Antonia M. Chávez 1.1 Examen 1 (11 de noviembre de 2019)	7	
2	Exámenes del grupo 2 María J. Hidalgo	11	
	2.1 Examen 1 (06 de noviembre de 2019)	11	
3	Exámenes del grupo 3 Francisco J. Martín	17	
	3.1 Examen 1 (28 de octubre de 2019	17	
4	Exámenes del grupo 4 José A. Alonso	23	
	4.1 Examen 1 (30 de octubre de 2019)	23	
5	Exámenes del grupo 5 Andrés Cordón y Miguel A. Martínez	29	
	5.1 Examen 1 (06 de noviembre de 2019)	29	
A	Resumen de funciones predefinidas de Haskell A.1 Resumen de funciones sobre TAD en Haskell	35 37	
В	Método de Pólya para la resolución de problemas B.1 Método de Pólya para la resolución de problemas matemáticos B.2 Método de Pólya para resolver problemas de programación		
Bi	ibliografía	45	

Índice general

Introducción

Este libro es una recopilación de las soluciones de ejercicios de los exámenes de programación funcional con Haskell de la asignatura de Informática (curso 2019-20) del Grado en Matemática de la Universidad de Sevilla.

Los exámenes se realizaron en el aula de informática y su duración fue de 2 horas. La materia de cada examen es la impartida desde el comienzo del curso (generalmente, el 1 de octubre) hasta la fecha del examen. Dicha materia se encuentra en los libros de temas y ejercicios del curso:

- Temas de programación funcional (curso 2019–20) ¹
- Ejercicios de "Informática de 1º de Matemáticas" (2019–20) ²
- Piensa en Haskell (Ejercicios de programación funcional con Haskell)

El libro consta de 5 capítulos correspondientes a 5 grupos de la asignatura. En cada capítulo hay una sección por cada uno de los exámenes del grupo. Los ejercicios de cada examen han sido propuestos por los profesores de su grupo (cuyos nombres aparecen en el título del capítulo). Sin embargo, los he modificado para unificar el estilo de su presentación.

Finalmente, el libro contiene dos apéndices. Uno con el método de Polya de resolución de problemas (sobre el que se hace énfasis durante todo el curso) y el otro con un resumen de las funciones de Haskell de uso más frecuente.

Los códigos del libro están disponibles en GitHub ⁴

Este libro es el 11º volumen de la serie de recopilaciones de exámenes de programación funcional con Haskell. Los volúmenes anteriores son

 Exámenes de "Programación funcional con Haskell". Vol. 1 (Curso 2009-10) ⁵

https://www.cs.us.es/~jalonso/cursos/ilm-19/temas/2019-20-IM-temas-PF.pdf

²https://www.cs.us.es/~jalonso/cursos/ilm-19/ejercicios/ejercicios-I1M-2019.pdf

³http://www.cs.us.es/~jalonso/publicaciones/Piensa en Haskell.pdf

⁴https://github.com/jaalonso/Examenes de PF con Haskell Vol11

⁵https://github.com/jaalonso/Examenes_de_PF_con_Haskell_Vol1

6 Índice general

 Exámenes de "Programación funcional con Haskell". Vol. 2 (Curso 2010-11) ⁶

- Exámenes de "Programación funcional con Haskell". Vol. 3 (Curso 2011– 12) ⁷
- Exámenes de "Programación funcional con Haskell". Vol. 4 (Curso 2012-13) 8
- Exámenes de "Programación funcional con Haskell". Vol. 5 (Curso 2013-14)
- Exámenes de "Programación funcional con Haskell". Vol. 6 (Curso 2014-15) 10
- Exámenes de "Programación funcional con Haskell". Vol. 6 (Curso 2015-16) 11
- Exámenes de "Programación funcional con Haskell". Vol. 8 (Curso 2016-17) 12
- Exámenes de "Programación funcional con Haskell". Vol. 9 (Curso 2017-18) 13
- Exámenes de "Programación funcional con Haskell". Vol. 10 (Curso 2018-19) 14

José A. Alonso Sevilla, 31 de octubre de 2019

⁶https://github.com/jaalonso/Examenes_de_PF_con_Haskell_Vol2
7https://github.com/jaalonso/Examenes_de_PF_con_Haskell_Vol3
8https://github.com/jaalonso/Examenes_de_PF_con_Haskell_Vol4
9https://github.com/jaalonso/Examenes_de_PF_con_Haskell_Vol5
10https://github.com/jaalonso/Examenes_de_PF_con_Haskell_Vol6
11https://github.com/jaalonso/Examenes_de_PF_con_Haskell_Vol7
12https://github.com/jaalonso/Examenes_de_PF_con_Haskell_Vol8
13https://github.com/jaalonso/Examenes_de_PF_con_Haskell_Vol9
14https://github.com/jaalonso/Examenes_de_PF_con_Haskell_Vol9

Exámenes del grupo 1

Antonia M. Chávez

1.1. Examen 1 (11 de noviembre de 2019)

```
-- Informática (1º del Grado en Matemáticas, Grupo 1)
-- 1º examen de evaluación continua (11 de noviembre de 2019)
import Data.List
-- Ejercicio 1. Definir la función de3en3 tal que (de3en3 x) es la lista
-- de dígitos de x agrupados de 3 en 3 consecutivamente. Por ejemplo,
     de3en3 12
                   == []
     de3en3 123
                  == [[1,2,3]]
                   == [[1,2,3],[2,3,4]]
    de3en3 1234
    de3en3 \ 12345 == [[1,2,3],[2,3,4],[3,4,5]]
    de3en3 \ 123450 == [[1,2,3],[2,3,4],[3,4,5],[4,5,0]]
     de3en3 \ 1234500 == [[1,2,3],[2,3,4],[3,4,5],[4,5,0],[5,0,0]]
-- 1ª solución
-- =========
de3en3 :: Integer -> [[Integer]]
de3en3 x = aux (digitos x)
 where aux (a:b:c:xs) = [a,b,c] : aux (b:c:xs)
       aux
                      = []
```

```
-- (digitos x) es la lista de los dígitos de x. Por ejemplo,
      digitos 325 == [3,2,5]
digitos :: Integer -> [Integer]
digitos x = [read [y] | y < - show x]
-- 2ª solución
-- =========
\frac{de3en3_2 x = [[a,b,c] | (a,(b,c)) \leftarrow zip xs (zip (tail xs) (tail (tail xs)))]}{de3en3_2 x = [[a,b,c] | (a,(b,c)) \leftarrow zip xs (zip (tail xs) (tail (tail xs)))]}
  where xs = digitos x
-- 3ª solución
-- ========
de3en3_3 x = [[a,b,c] | (a,b,c) <- zip3 xs (tail xs) (drop 2 xs)]
  where xs = digitos x
-- Ejercicio 2. Un número es endiablado si su doble contiene al 666. Por
-- ejemplo, el número 1333 es endiablado ya que su doble 2666 contiene
-- al 666.
-- Definir la función esEndiablado tal que (esEndiablado x) se verifica
-- si x es endiablado. Por ejemplo,
    esEndiablado 1333 == True
     esEndiablado 23331 == True
     esEndiablado 28331114 == True
     esEndiablado 3339 == False
-- 1ª solución
esEndiablado :: Integer -> Bool
esEndiablado x = [6,6,6] `elem` de3en3 (2*x)
-- 2ª solución
esEndiablado2 :: Integer -> Bool
esEndiablado2 x = "666" `isInfixOf` show (2*x)
```

```
-- Ejercicio 3. Calcular la lista de los 3 primeros números endiablados
-- mayores que 456.
-- El cálculo es
   \lambda> take 3 [x | x <- [457 ..], esEndiablado x]
-- [833, 1333, 1833]
-- Ejercicio 4. Definir la función
-- entreMedias :: [Float] -> [Float]
-- tal que (entreMedias xs) es la lista obtenida intercalando entre cada
-- dos elementos consecutivos de xs su media aritmética. Por ejemplo,
-- entreMedias [2,7,1,8] = [2.0,4.5,7.0,4.0,1.0,4.5,8.0]
    entreMedias [2] == [2.0]
    entreMedias []
                         == []
entreMedias :: [Float] -> [Float]
entreMedias [] = []
entreMedias [x] = [x]
entreMedias (x:y:xs) = x: media x y : entreMedias (y:xs)
 where media x y = (x+y)/2
-- Ejercicio 5.1. La serie de potencias 1 + x + x^2/2! + x^3/3! + ...
-- define la función exponencial e^x que es exp x en Haskell.
-- Para calcular e^3 bastará calcular la serie para x=3
-- Definir la función serieExp tal que (serieExp x n) es el valor
-- aproximado de e^x que resulta de sumar n términos de la serie
-- comenzando en 0. Por ejemplo,
  serieExp \ 3 \ 5 = 1 + 3 + 3^2/2! + 3^3/3! + 3^4/4!
               = 16.375
serieExp :: Integer -> Integer -> Double
serieExp x m =
 sum [fromIntegral(x^n)/fromIntegral(fact n) | n <- [0 ..m-1]]
```

```
-- (fact n) es el factorial de n. Por ejemplo,
     fact 3 == 6
fact :: Integer -> Integer
fact n = product [1 ... n]
-- Ejercicio 5.2. Definir la función errorExp tal que (errorExp n) es
-- el errorExp que se comente en el cálculo de e^3 si tenemos en cuenta
-- sólo n términos de la serie. Por ejemplo,
     errorExp 27 == 7.105427357601002e-15
     errorExp 10 == 2.2144066044813115e-2
errorExp :: Integer -> Double
errorExp n = abs (exp 3 - serieExp 3 n)
-- Ejercicio 5.3. Definir la funcion minimoA tal que (minimoA x) es el
-- menor número de términos tal que el errorExp en el calculo de e^3 con
-- n términos no supera a x. Por ejemplo, el errorExp que cometemos
-- calculando e^3 con la suma de 6 términos es
     errorExp 6 == 1.6855369231876693
-- y el menor n tal que el errorExp no es mayor que 2 es 6
-- minimoA 2 == 6
-- ya que con 5 términos el errorExp es (errorExp 5 = 3.710536923187668)
minimoA :: Double -> Integer
minimoA x = head [n \mid n < [0 ..], errorExp n <= x]
```

Exámenes del grupo 2

María J. Hidalgo

2.1. Examen 1 (06 de noviembre de 2019)

```
sumaS :: Double -> Double
sumaS n = sum [1/(k*(k+1)) | k < - [1..n]]
-- Ejercicio 1.2. Definir la función
     errorS :: Double -> Double
-- tal que (errorS x) es el menor número de términos de la serie
-- necesarios para obtener su límite con un error menor que x.
-- Por ejemplo,
    errorS 0.01 == 100.0
    errorS 0.001 == 999.0
errorS :: Double -> Double
errorS x = head [m \mid m \leftarrow [1..], abs ((sumaS m) - 1) < x]
-- Ejercicio 2. ¿Cuántos números de tres dígitos tienen al menos un
-- dígito par?
__ ______
solucion :: Int
solucion = length [n | n <- [100..999], tieneAlgunDP n]
-- (tieneAlgunDP n) se verifica si n tiene agún dígito par. Por ejemplo,
     tieneAlgunDP 254 == True
     tieneAlgunDP 35 == False
tieneAlgunDP :: Int -> Bool
tieneAlgunDP n = or [even x \mid x \leftarrow digitos n]
-- (digitos n) es la lista de los dígitos de n. Por ejemplo,
     digitos 325 == [3,2,5]
digitos :: Int -> [Int]
digitos n = [read [x] | x <- show n]
-- El cálculo es
-- λ> solucion
-- 775
-- 2ª solución
```

```
-- ========
solucion2 :: Int
solucion2 = length [n | n <- [100..999], tieneAlgunDP2 n]
-- (tieneAlgunDP2 n) se verifica si n tiene agún dígito par. Por ejemplo,
     tieneAlgunDP2 254 == True
     tieneAlgunDP2 35 == False
tieneAlgunDP2 :: Int -> Bool
tieneAlgunDP2 n = not (null (intersect (show n) "02468"))
-- Ejercicio 3.1. Decimos que un número n es curioso si la suma de sus
-- dígitos termina igual que el producto de los mismos.
-- Definir la función
    esCurioso :: Int -> Bool
-- tal que (esCurioso n) se verifica si n es un número curioso. Por
-- ejemplo,
   esCurioso 109 = True
-- esCurioso 121 = False
-- 1ª solución
-- =========
esCurioso :: Int -> Bool
esCurioso n = x `rem` 10 == y `rem` 10
 where x = sumaDigitos n
       y = prodDigitos n
-- (sumaDigitos n) es la suma de los dígitos de n. Por ejemplo,
     sumaDigitos 325 == 10
sumaDigitos :: Int -> Int
sumaDigitos n | n < 10
             otherwise = (n `rem` 10) + sumaDigitos (n `div` 10)
-- (prodDigitos n) es el producto de los dígitos de n. Por ejemplo,
     prodDigitos 325 == 30
prodDigitos :: Int -> Int
```

```
prodDigitos n | n < 10 = n
              | otherwise = (n `rem` 10) * prodDigitos (n `div` 10)
-- 2ª solución
-- =========
esCurioso2 :: Int -> Bool
esCurioso2 n = x `rem` 10 == y `rem` 10
 where x = sum (digitos n)
        y = product (digitos n)
-- Equivalencia
prop_esCuriosoEquiv :: Int -> Property
prop esCuriosoEquiv n =
  n >= 0 ==> esCurioso n == esCurioso2 n
-- La comprobación es
    λ> quickCheck prop_esCuriosoEquiv
    +++ OK, passed 100 tests.
-- Ejercicio 3.2. Definir la función
    curiososHasta :: Int -> [Int]
-- tal que (curiososHasta n) es la lista de los números curiosos menores
-- o iguales que n. Por ejemplo,
     curiososHasta\ 125 == [1,2,3,4,5,6,7,8,9,22,48,84,109,123]
curiososHasta :: Int -> [Int]
curiososHasta n = [x \mid x \leftarrow [1..n], esCurioso x]
-- Ejercicio 3.3. Comprobar con QuickCheck la siguiente propiedad: un
-- número n curioso si, y sólo si, lo es el número obtenido al invertir
-- sus dígitos.
-- La propiedad es
propCurioso :: Int -> Property
propCurioso n =
```

```
n >= 0 ==> esCurioso n ==> esCurioso (read (reverse (show n)))
-- La comprobación es
    λ> quickCheck propCurioso
     +++ OK, passed 100 tests.
-- Ejercicio 4.1. Definir la función
     mayorPre :: Eq a \Rightarrow [a] -> [a]
-- tal que (mayorPre xs ys) calcula el mayor prefijo común a xs e ys. Por
-- ejemplo,
     mayorPre "masa" "madre"
                                  == "ma"
                                  == ""
     mayorPre "masa" "padre"
    mayorPre "hola" "hielo" == "h"
     mayorPre "helado" "heladeria" == "helad"
mayorPre :: Eq a => [a] -> [a] -> [a]
mayorPre [] = []
mayorPre [] _ = []
mayorPre (x:xs) (y:ys) \mid x == y = x : mayorPre xs ys
                      | otherwise = []
-- Ejercicio 4.2. Definir la función
     mayorSuf :: Eq a => [a] -> [a]
-- tal que (mayorSuf xs ys) calcula el mayor sufijo común a xs e ys. Por
-- ejemplo,
     mayorSuf "masa" "madre"
     mayorSuf "masa" "madera"
                                    == "a"
     mayorSuf "masa" "mesa"
     mayorSuf "panaderia" "heladeria" == "aderia"
mayorSuf :: Eq a => [a] -> [a] -> [a]
mayorSuf xs ys = reverse (mayorPre (reverse xs) (reverse ys))
```

Exámenes del grupo 3

Francisco J. Martín

3.1. Examen 1 (28 de octubre de 2019

```
-- Informática (1º del Grado en Matemáticas, Grupo 3)
-- 1º examen de evaluación continua (28 de octubre de 2019)
  -- Ejercicio 1. Supongamos un tablero de ajedrez infinito de casillas de
-- tamaño 1x1, con una casilla negra situada entre las posiciones (0,0),
-- (1,0), (0,1) y (1,1). El tablero se extiende en todas direcciones
-- abarcando todo el plano. De esta forma, cada punto del plano está
-- pintado de color negro (el interior de las casillas negras y los
-- bordes) o de color blanco (el interior de las casillas blancas). Por
-- ejemplo, el punto (0.5,0.5) se encuentra pintado de negro (en el
-- interior de una casilla negra), el punto (0.5,1) también se encuentra
-- pintado de negro (en el borde de las casillas) y el punto (0.5,1.5)
-- se encuentra pintado de blanco (en el interior de una casilla blanca).
-- Definir la función
    posicionNegra :: (Double, Double) -> Bool
-- tal que (posicionNegra (x,y)) se verifica si el punto (x,y) se encuentra
-- pintado de negro en el tablero de ajedrez infinito. Por ejemplo,
   posicionNegra (0.5, 0.5) == True
    posicionNegra (0.5,1)
                            == True
  posicionNegra (0.5,1.5)
                            == False
   posicionNegra (-0.5,0.5) == False
```

```
posicionNegra (0.5, -0.5) == False
    posicionNegra (-0.5, -0.5) == True
-- 1ª solución
posicionNegra :: (Double, Double) -> Bool
posicionNegra (x,y)
 | floor x == ceiling x = True
  | floor y == ceiling y = True
  | even (floor x + floor y) = True
  | otherwise = False
-- 2ª solución
posicionNegra2 :: (Double, Double) -> Bool
posicionNegra2(x,y) =
     floor x == ceiling x
  || floor y == ceiling y
  || even (floor x + floor y)
-- Ejercicio 2. Una lista de números [a(1),a(2),...] se denomina
-- posicionalmente coprima si para cada i ∈ {1, 2,...} se cumple que i y
-- a(i) son coprimos, es decir, el máximo común divisor de a(i) e i es
-- 1. Por ejemplo, la lista [3,5,7] es posicionalmente coprima pues 3 es
-- coprimo con 1, 5 es coprimo con 2 y 7 es coprimo con 3; por otro
-- lado, la lista [2,4,6] no es posicionalmente coprima pues 4 no es
-- coprimo con 2, ni 6 es coprimo con 3.
-- Definir la función
      posicionalmenteCoprima :: [Integer] -> Bool
-- tal que (posicionalmenteCoprima xs) se cumple si la lista xs es
-- posicionalmente coprima. Por ejemplo,
     posicionalmenteCoprima [3,5,7] == True
     posicionalmenteCoprima [2,4,6] == False
     posicionalmenteCoprima []
                                    == True
     posicionalmenteCoprima [6]
                                    == True
-- 1ª solución
posicionalmenteCoprima :: [Integer] -> Bool
```

```
posicionalmenteCoprima xs =
  and [\gcd x i == 1 \mid (x,i) <- zip xs [1..]]
-- 2ª solución
posicionalmenteCoprima2 :: [Integer] -> Bool
posicionalmenteCoprima2 xs = aux xs [1..]
 where aux []
                        = True
       aux (x:xs) (n:ns) = gcd x n == 1 \&\& aux xs ns
-- Ejercicio 3. Una forma de aproximar el valor del número \pi es usando
-- la siguiente igualdad:
           \pi 4 16 36 64
           --- = --- * ---- * ---- * ...
            2 3 15 35 63
-- Es decir, el producto de los términos de la secuencia cuyo término general
-- n-ésimo es:
                    4*k^2
            s(k) = -----, para k > 0
                   4*k^2-1
-- Definir la función
     aproximaPi :: Double -> Double
-- tal que (aproximaPi n) es la aproximación del número \pi calculada con la
-- serie anterior hasta el término n-ésimo (contando desde 0). Por ejemplo,
     aproximaPi 10 == 3.0677038066434985
     aproximaPi 100 == 3.133787490628162
     aproximaPi 1000 == 3.140807746030402
-- 1ª definición
aproximaPi :: Double -> Double
aproximaPi n =
 2 * product [(4*k^2) / (4*k^2-1) | k <- [1..n]]
-- 2ª definición
aproximaPiR :: Double -> Double
aproximaPiR 0 = 2
```

```
aproximaPiR n = aproximaPiR (n-1) * 4 * n^2 / (4*n^2-1)
-- Ejercicio 4. La mecánica habitual para sumar números naturales
-- consiste en sumar todas las cifras que ocupan la misma posición,
-- tomadas en orden desde las unidades, acumulando una 'llevada' desde
-- una posición a la siguiente. Consideremos un proceso de 'suma
-- sesgada' que no acumula la llevada, por ejemplo la suma ordinaria de
-- 1357 y 579 es 1936, sin embargo el valor de la suma sesgada es 1826:
         Suma con llevada:
                                         Suma sesgada:
              1 1
            1 3 5 7
                                             1 3 5 7
         + 579
                                         + 579
            1 9 3 6
                                             1 8 2 6
-- Definir la función:
     sumaSesgada :: Integer -> Integer
-- tal que (sumaSesgada n m) es la 'suma sesgada' de los números naturales
-- n y m. Por ejemplo,
     sumaSesgada 1357 579 == 1826
     sumaSesgada 9999 999 == 9888
    sumaSesgada 0 1234 == 1234
-- 1ª solución
-- =========
sumaSesgada :: Integer -> Integer
sumaSesgada 0 m = m
sumaSesgada n 0 = n
sumaSesgada n m =
  sumaSesgada (div n 10) (div m 10) * 10 + mod (mod n 10 + mod m 10) 10
-- 2ª solución
-- ========
sumaSesgada2 :: Integer -> Integer
sumaSesgada2 n m =
```

Exámenes del grupo 4

José A. Alonso

4.1. Examen 1 (30 de octubre de 2019)

```
-- Informática (1º del Grado en Matemáticas, Grupo 4)
-- 1º examen de evaluación continua (30 de octubre de 2019)
-- Nota: La puntuación de cada ejercicio es 2.5 puntos.
import Test.QuickCheck
-- Ejercicio 1. La distancia de Hamming entre dos listas es el número de
-- posiciones en que los correspondientes elementos son distintos. Por
-- ejemplo, la distancia de Hamming entre "roma" y "loba" es 2 (porque
-- hay 2 posiciones en las que los elementos correspondientes son
-- distintos: la 1º y la 3º).
-- Definir la función
     distancia :: Eq a => [a] -> [a] -> Int
-- tal que (distancia xs ys) es la distancia de Hamming entre xs e
-- ys. Por ejemplo,
-- distancia "romano" "comino" == 2
     distancia "romano" "camino" == 3
    distancia "roma" "comino" == 2
    distancia "roma" "camino" == 3
    distancia "romano" "ron" == 1
```

```
distancia "romano" "cama"
     distancia "romano" "rama" == 1
-- 1ª definición (por comprensión):
distancia :: Eq a => [a] -> [a] -> Int
distancia xs ys = sum [1 \mid (x,y) \leftarrow zip xs ys, x /= y]
-- 2ª definición (por recursión):
distancia2 :: Eq a => [a] -> [a] -> Int
distancia2 [] = 0
distancia2 [] = 0
distancia2 (x:xs) (y:ys) | x /= y = 1 + distancia2 xs ys
                        | otherwise = distancia2 xs ys
-- Ejercicio 2. Definir la función
     esPotencia :: Integer -> Integer -> Bool
-- tal que (esPotencia x a) se verifica si x es una potencia de a. Por
-- ejemplo,
    esPotencia 32 2 == True
    esPotencia 42 2 == False
-- 1ª definición (por comprensión):
esPotencia :: Integer -> Integer -> Bool
esPotencia x a = x 'elem' [a^n | n <- [0..x]]
-- 2ª definición (por recursión):
esPotencia2 :: Integer -> Integer -> Bool
esPotencia2 x a = aux x a 0
 where aux x a b \mid b > x
                           = False
                  | otherwise = x == a ^b | aux x a (b+1)
-- 3ª definición (por recursión):
esPotencia3 :: Integer -> Integer -> Bool
esPotencia3 0 = False
esPotencia3 1 a = True
esPotencia3 _ 1 = False
esPotencia3 x a = rem x a == 0 \&\& esPotencia3 (div x a) a
```

```
-- La propiedad de equivalencia es
prop_equiv_esPotencia :: Integer -> Integer -> Property
prop equiv esPotencia x a =
    x > 0 \& \& a > 0 ==>
    esPotencia2 x a == b &&
    esPotencia3 x a == b
    where b = esPotencia \times a
-- La comprobación es
     ghci> quickCheck prop_equiv_esPotencia
     +++ OK, passed 100 tests.
-- Ejercicio 3.1. Definir la función
      sumaListas :: [Int] -> [Int] -> [Int]
-- tal que (sumaListas xs ys) es la suma de las elementos
-- correspondientes de las lista xs e ys. Por ejemplo,
      sumaListas [2,3,4] [1,2,5] == [3,5,9]
     sumaListas [2,3,4] [1,2] == [3,5,4]
    sumaListas [2,3] [1,2,5] == [3,5,5]
-- 1ª definición
sumaListas :: [Int] -> [Int] -> [Int]
sumaListas [] ys
                       = ys
sumaListas xs []
                         = XS
sumaListas (x:xs) (y:ys) = x+y : sumaListas xs ys
-- 2ª definición
sumaListas2 :: [Int] -> [Int] -> [Int]
sumaListas2 xs ys = [x+y \mid (x,y) \leftarrow zip (xs ++ replicate (k-m) 0)
                                        (ys ++ replicate (k-n) 0)]
 where m = length xs
        n = length vs
        k = max m n
-- Ejercicio 3.2. Comprobar con QuickCheck que el número de elementos de
-- (sumaListas xs ys) es el máximo de los números de elementos de xs e ys.
```

```
-- La propiedad es
prop_sumaListas :: [Int] -> [Int] -> Bool
prop_sumaListas xs ys =
  length (sumaListas xs ys) == max (length xs) (length ys)
-- La comprobación es
     λ> quickCheck prop sumaListas
     +++ OK, passed 100 tests.
-- Ejercicio 4. Un número primo equilibrado es un número primo que es la
-- media aritmética de su primo anterior y siguiente. Por ejemplo, 5 es
-- un primo equilibrado porque es la media de 3 y 7; pero 7 no lo es
-- porque no es la media de 5 y 11.
-- Definir la función
      primosEquilibrados :: Int -> [Int]
-- tal que (primosEquilibrados n) es la lista de los números primos
-- equilibrados menores o iguales que n. Por ejemplo,
     ghci> primosEquilibrados 1000
      [5,53,157,173,211,257,263,373,563,593,607,653,733,947,977]
-- 1ª solución
- - =========
primosEquilibrados :: Int -> [Int]
primosEquilibrados n =
  [x \mid x \leftarrow [5,7..n]
     , esPrimo x
     , esEquilibrado x]
-- (esPrimo n) se verifica si n es primo. Por ejemplo,
      esPrimo 5 == True
      esPrimo 6 == False
esPrimo :: Int -> Bool
esPrimo n = divisores n == [1,n]
```

```
-- (primos n) es la lista de los números primos menores o iguales que
-- n. Por ejemplo,
     primos 50 == [2,3,5,7,11,13,17,19,23,29,31,37,41,43,47]
primos :: Int -> [Int]
primos n = [x \mid x \leftarrow [1..n], esPrimo x]
-- (divisores n) es la lista de los divisores de n Por ejemplo,
      divisores 36 = [1,2,3,4,6,9,12,18,36]
divisores :: Int -> [Int]
divisores n = [x \mid x \leftarrow [1..n], n \mod x == 0]
-- (esEquilibrado x) se verifica si x es la media de su primo anterior y
-- su primo siguiente. Por ejemplo,
     esEquilibrado 5 == True
      esEquilibrado 7 == False
esEquilibrado :: Int -> Bool
esEquilibrado x = 2 * x == primoAnterior x + primoSiguiente x
-- (primoAnterior x) es el primo anterior a x. Por ejemplo,
      primoAnterior 22 == 19
primoAnterior :: Int -> Int
primoAnterior x = head [y \mid y \leftarrow [x-1,x-2..]
                           , esPrimo y]
-- (primoSiguiente x) es el primo siguiente a x. Por ejemplo,
      primoSiguiente 22 == 23
      primoSiguiente 23 == 29
primoSiguiente :: Int -> Int
primoSiguiente x = head [y \mid y \leftarrow [x+1..]]
                           , esPrimo y]
-- 2ª solución
- - ========
primosEquilibrados2 :: Int -> [Int]
primosEquilibrados2 n = aux (primos n)
 where aux (x1:x2:x3:xs)
          | 2*x2 == x1+x3 = x2 : aux (x2:x3:xs)
          | otherwise = aux (x2:x3:xs)
                          = []
        aux _
```

Exámenes del grupo 5

Andrés Cordón y Miguel A. Martínez

5.1. Examen 1 (06 de noviembre de 2019)

```
-- Informática (1º del Grado en Matemáticas), Grupo 5
-- 1º examen de evaluación continua (6 de noviembre de 2019)
-- Ejercicio 1. Definir la función
      numerosLongitudDivisores :: Int -> Int -> [Int]
-- tal que (numerosLongitudDivisores n k) es la lista de los números de
-- n dígitos que ienen k o más divisores. Por ejemplo:
      numerosLongitudDivisores 3 25 == [720,840,900,960]
      numerosLongitudDivisores 2 10 == [48,60,72,80,84,90,96]
numerosLongitudDivisores :: Int -> Int -> [Int]
numerosLongitudDivisores n k =
  [x \mid x \leftarrow [10^{n-1}...10^{n-1}], length (divisores x) >= k]
-- (divisores x) es la lista de los divisores de x. Por ejemplo,
      divisores 60 == [1,2,3,4,5,6,10,12,15,20,30,60]
divisores :: Int -> [Int]
divisores x = [z \mid z \leftarrow [1..x], mod x z == 0]
-- Ejercicio 2.1. Se consideran listas de pares con información sobre
```

```
-- las notas obtenidas por alumnos en un examen. Por ejemplo:
     notas :: [(String,Float)]
     notas = [("Juan",5.6),("Ana",9.5),("Pedro",4),("Luisa",9.5),
               ("Mateo",9),("Isabel",3.5)]
-- Definir la función
      candidatos :: [(String,Float)] -> [String]
-- tal que (candidatos xs) es la lista de los nombres de los alumnos con
-- la nota máxima en la lista de notas xs. Por ejemplo,
     candidatos lista = ["Ana","Luisa"]
notas :: [(String,Float)]
notas = [("Juan",5.6),("Ana",9.5),("Pedro",4),("Luisa",9.5),
         ("Mateo",9),("Isabel",3.5)]
candidatos :: [(String,Float)] -> [String]
candidatos xs = [cs \mid (cs,n) \leftarrow xs, n == k]
 where k = maximum [z | (,z) <- xs]
-- Ejercicio 2.2. Definir la función
     media :: [(String,Float)] -> Float
-- tal que (media xs) es la nota media correspondiente a la lista de
-- notas xs. Por ejemplo,
     media notas == 6.85
media :: [(String,Float)] -> Float
media xs = sum [z | (\_,z) \leftarrow xs] / fromIntegral (length xs)
-- Ejercicio 3.1. Un número capicúa es aquel que se lee igual de
-- izquierda a derecha que de derecha a izquierda. Una forma de
-- consequir capicúas a partir de un número entero x es sumarle su
-- reverso (x con las cifras invertidas) y repetir el proceso hasta
-- conseguir un número capicúa. Por ejemplo:
-- + para el 15 tenemos:
        15 + 51 = 66
-- + para el 87 tenemos:
```

```
87 + 78 = 165,
          165 + 561 = 726,
          726 + 627 = 1353,
        1353 + 3531 = 4884
-- Definir la función
     unPaso :: Integer -> Integer
-- talque (unPaso x) es el entero obtenido al sumar x y su reverso. Por
-- ejemplo,
     unPaso 87 == 165
     unPaso 121 == 242
unPaso :: Integer -> Integer
unPaso x = x + reverso x
-- (reverso x) es el número obtenido al invertir los dígitos de x. Por
-- ejemplo,
     reverso 325 == 523
reverso :: Integer -> Integer
reverso x = read (reverse (show x))
-- Ejercicio 3.2. Definir la función
     hastaCapicua:: Integer -> Integer
-- tal que (hastaCapicua x) es el número de veces que debemos repetir
-- el proceso anterior hasta conseguir un número capicúa. Por ejemplo,
     hastaCapicua 15
     hastaCapicua 87 == 4
     hastaCapicua 1221 == 0
hastaCapicua :: Integer -> Integer
hastaCapicua x
  \mid esCapicua x = 0
  | otherwise = 1 + hastaCapicua (unPaso x)
-- (esCapicua x) se verifica i x es un número capicúa. Por ejemplo,
     esCapicua 32523 == True
     esCapicua 32532 == False
```

```
esCapicua :: Integer -> Bool
esCapicua x = x == reverso x
-- Ejercicio 4. Se dice que una lista de números zigzaguea si el primer
-- elemento de la lista es (estrictamente) menor que el segundo, el
-- segundo es mayor que el tercero, el tercero es menor que el cuarto,
-- el cuarto es mayor que el quinto y así sucesivamente. (Si la lista
-- tiene menos de dos elementos, entonces asumimos que zigzaguea).
-- Definir el predicado
     zigzaguea :: (Num a,Ord a) -> [a] -> Bool
-- tal que (zigzaguea xs) se verifica si la lista xs zigzaguea. Por
-- ejemplo,
   zigzaguea [3]
                             == True
    zigzaguea [4,3] == False
-- zigzaguea [2,4,1,3,-1,5] == True
-- zigzaguea [1,2,2,-3,7] == False
-- 1º solución (por recursión)
zigzagueal :: (Num a,Ord a) => [a] -> Bool
zigzagueal (x:y:z:xs) = x < y \& y > z \& x zigzagueal (z:xs)
zigzaguea1 [x,y] = x < y
                     = True
zigzaguea1
-- 2ª solución (por comprensión)
zigzaguea2 :: (Num a,Ord a) => [a] -> Bool
zigzaguea2 xs
  | length xs < 2 = True
  | otherwise = and [x < y | (x,y) \leftarrow posImpares ys]
                && and [x > y \mid (x,y) \leftarrow posPares ys]
    where ys = zip xs (tail xs)
-- (posImpares xs) es la lista de los elementos de xs en las posiciones
-- impares (empezando a contar en 1). Por ejemplo,
     posImpares [2,4,1,3,-1,5] == [2,1,-1]
posImpares :: [a] -> [a]
posImpares xs = [x \mid (x,n) \leftarrow zip xs [1..], odd n]
```

```
-- (posPares xs) es la lista de los elementos de xs en las posiciones
-- pares (empezando a contar en 1). Por ejemplo,
     posPares [2,4,1,3,-1,5] == [4,3,5]
posPares :: [a] -> [a]
posPares xs = [x \mid (x,n) \leftarrow zip xs [1..], even n]
-- 3ª solución (por comprensión)
zigzaguea3 :: (Num a,Ord a) => [a] -> Bool
zigzaguea3 xs =
  (length xs < 2)
  || (and [x < y | (x,y) \leftarrow paresImpares xs] \&\&
      and [x > y \mid (x,y) \leftarrow imparesPares xs])
-- (paresImpares xs) es la lista de los pares formado por los elementos
-- de xs que están en las posiciones pares junto con su siguiente
-- elemento. Por ejemplo,
      paresImpares [0..10] == [(0,1),(2,3),(4,5),(6,7),(8,9)]
paresImpares :: [a] -> [(a,a)]
paresImpares (x:y:zs) = (x,y) : paresImpares zs
paresImpares _ = []
-- (imparesPares xs) es la lista de los pares formado por los elementos
-- de xs que están en las posiciones impares junto con su siguiente
-- elemento. Por ejemplo,
      imparesPares [0..10] == [(1,2),(3,4),(5,6),(7,8),(9,10)]
imparesPares :: [a] -> [(a,a)]
imparesPares (_:x:y:zs) = (x,y) : imparesPares (y:zs)
imparesPares _
                 = []
```

Apéndice A

Resumen de funciones predefinidas de Haskell

```
1. x + y es la suma de x e y.
 2. |x - y| es la resta de x e y.
 3. x / y es el cociente de x entre y.
 4.
     \mathbf{x} \hat{\mathbf{y}} es x elevado a y.
 5.
     x == y se verifica si x es igual a y.
     x \neq y se verifica si x es distinto de y.
 6.
 7.
     x < y | se verifica si x es menor que y.
 8.
     x \leftarrow y se verifica si x es menor o igual que y.
     x > y | se verifica si x es mayor que y.
 9.
10.
     x >= y | se verifica si x es mayor o igual que y.
11.
     x \& y es la conjunción de x e y.
12.
     x | | y es la disyunción de x e y.
13.
     x:ys es la lista obtenida añadiendo x al principio de ys.
14.
     xs ++ ys es la concatenación de xs e ys.
     xs !! n es el elemento n-ésimo de xs.
15.
     f . g es la composición de f y g.
16.
17.
     abs x es el valor absoluto de x.
     and xs es la conjunción de la lista de booleanos xs.
18.
19.
     ceiling x es el menor entero no menor que x.
20.
     chr n es el carácter cuyo código ASCII es n.
     concat xss es la concatenación de la lista de listas xss.
21.
22.
     const x y es x.
```

- 23. curry f es la versión curryficada de la función f.
- 24. div x y es la división entera de x entre y.
- 25. drop n xs borra los n primeros elementos de xs.
- 26. dropWhile p xs borra el mayor prefijo de xs cuyos elementos satisfacen el predicado p.
- 27. $\begin{vmatrix} elem \times ys \end{vmatrix}$ se verifica si x pertenece a ys.
- 29. filter p xs es la lista de elementos de la lista xs que verifican el predicado p.
- 30. | flip f x y | es f y x.
- 31. floor x es el mayor entero no mayor que x.
- 32. foldl f e xs pliega xs de izquierda a derecha usando el operador f y el valor inicial e.
- 33. foldr f e xs pliega xs de derecha a izquierda usando el operador f y el valor inicial e.
- 34. fromIntegral x transforma el número entero x al tipo numérico correspondiente.
- 35. | fst p | es el primer elemento del par p.
- 36. $| gcd \times y |$ es el máximo común divisor de de x e y.
- 37. head xs es el primer elemento de la lista xs.
- 38. init xs es la lista obtenida eliminando el último elemento de xs.
- 39. iterate f x es la lista [x, f(x), f(f(x)), ...].
- 40. <u>last xs</u> es el último elemento de la lista xs.
- 41. length xs es el número de elementos de la lista xs.
- 42. map f xs es la lista obtenida aplicado f a cada elemento de xs.
- 43. max x y es el máximo de x e y.
- 44. maximum xs es el máximo elemento de la lista xs.
- 45. $min \times y$ es el mínimo de x e y.
- 46. minimum xs es el mínimo elemento de la lista xs.
- 47. $| mod \times y |$ es el resto de x entre y.
- 48. not x es la negación lógica del booleano x.
- 49. noElem x ys se verifica si x no pertenece a ys.
- 50. null xs se verifica si xs es la lista vacía.
- 51. $odd \times se$ verifica si x es impar.
- 52. or xs es la disyunción de la lista de booleanos xs.
- 53. ord c es el código ASCII del carácter c.

- 54. product xs es el producto de la lista de números xs.
- 55. read c es la expresión representada por la cadena c.
- 56. rem x y es el resto de x entre y.
- 57. repeat x es la lista infinita [x, x, x, ...].
- 58. replicate n x es la lista formada por n veces el elemento x.
- 59. reverse xs es la inversa de la lista xs.
- 60. round x es el redondeo de x al entero más cercano.
- 61. scanr f e xs es la lista de los resultados de plegar xs por la derecha con f y e.
- 62. show x es la representación de x como cadena.
- 63. $\begin{vmatrix} signum x \end{vmatrix}$ es 1 si x es positivo, 0 si x es cero y -1 si x es negativo.
- 64. snd p es el segundo elemento del par p.
- 65. splitAt n xs es (take n xs, drop n xs).
- 66. sqrt x es la raíz cuadrada de x.
- 67. sum xs es la suma de la lista numérica xs.
- 68. | tail xs | es la lista obtenida eliminando el primer elemento de xs.
- 69. take n xs es la lista de los n primeros elementos de xs.
- 70. takeWhile p xs es el mayor prefijo de xs cuyos elementos satisfacen el predicado p.
- 71. uncurry f es la versión cartesiana de la función f.
- 72. | until p f x | aplica f a x hasta que se verifique p.
- 73. zip xs ys es la lista de pares formado por los correspondientes elementos de xs e ys.
- 74. zipWith f xs ys se obtiene aplicando f a los correspondientes elementos de xs e ys.

A.1. Resumen de funciones sobre TAD en Haskell

A.1.1. Polinomios

- 1. polCero es el polinomio cero.
- 2. (esPolCero p) se verifica si p es el polinomio cero.
- 3. (consPol n b p) es el polinomio $bx^n + p$.
- 4. (grado p) es el grado del polinomio p.

- 5. (coefLider p) es el coeficiente líder del polinomio p.
- 6. (restoPol p) es el resto del polinomio p.

A.1.2. Vectores y matrices (Data.Array)

- 1. (range m n) es la lista de los índices del m al n.
- 2. (index (m,n) i) es el ordinal del índice i en (m,n).
- 3. (inRange (m,n) i) se verifica si el índice i está dentro del rango limitado por m y n.
- 4. (rangeSize (m,n)) es el número de elementos en el rango limitado por m y n.
- 5. (array (1,n) [(i, f i) | i <- [1..n]) es el vector de dimensión n cuyo elemento i-ésimo es f i.
- 6. (array ((1,1),(m,n)) [((i,j), f i j) | i <- [1..m], j <- [1..n]]) es la matriz de dimensión m.n cuyo elemento (i,j)-ésimo es f i j.
- 7. (array (m,n) ivs) es la tabla de índices en el rango limitado por m y n definida por la lista de asociación ivs (cuyos elementos son pares de la forma (índice, valor)).
- 8. (t ! i) es el valor del índice i en la tabla t.
- 9. | (bounds t) | es el rango de la tabla t.
- 10. (indices t) es la lista de los índices de la tabla t.
- 11. (elems t) es la lista de los elementos de la tabla t.
- 12. (assocs t) es la lista de asociaciones de la tabla t.
- 13. (t // ivs) es la tabla t asignándole a los índices de la lista de asociación ivs sus correspondientes valores.
- 14. (listArray (m,n) vs) es la tabla cuyo rango es (m,n) y cuya lista de valores es vs.
- 15. (accumArray f v (m,n) ivs) es la tabla de rango (m,n) tal que el valor del índice i se obtiene acumulando la aplicación de la función f al valor inicial v y a los valores de la lista de asociación ivs cuyo índice es i.

A.1.3. Tablas

- 1. (tabla ivs) es la tabla correspondiente a la lista de asociación ivs (que es una lista de pares formados por los índices y los valores).
- 2. (valor t i) es el valor del índice i en la tabla t.
- 3. (modifica (i,v) t) es la tabla obtenida modificando en la tabla t el valor de i por v.

A.1.4. Grafos

- 1. (creaGrafo d cs as) es un grafo (dirigido o no, según el valor de o), con el par de cotas cs y listas de aristas as (cada arista es un trío formado por los dos vértices y su peso).
- 2. (dirigido g) se verifica si g es dirigido.
- 3. (nodos g) es la lista de todos los nodos del grafo g.
- 4. (aristas g) es la lista de las aristas del grafo g.
- 5. (adyacentes g v) es la lista de los vértices adyacentes al nodo v en el grafo g.
- 6. (aristaEn g a) se verifica si a es una arista del grafo g.
- 7. (peso v1 v2 g) es el peso de la arista que une los vértices v1 y v2 en el grafo g.

Apéndice B

Método de Pólya para la resolución de problemas

B.1. Método de Pólya para la resolución de problemas matemáticos

Para resolver un problema se necesita:

Paso 1: Entender el problema

- ¿Cuál es la incógnita?, ¿Cuáles son los datos?
- ¿Cuál es la condición? ¿Es la condición suficiente para determinar la incógnita? ¿Es insuficiente? ¿Redundante? ¿Contradictoria?

Paso 2: Configurar un plan

- ¿Te has encontrado con un problema semejante? ¿O has visto el mismo problema planteado en forma ligeramente diferente?
- ¿Conoces algún problema relacionado con éste? ¿Conoces algún teorema que te pueda ser útil? Mira atentamente la incógnita y trata de recordar un problema que sea familiar y que tenga la misma incógnita o una incógnita similar.
- He aquí un problema relacionado al tuyo y que ya has resuelto ya. ¿Puedes utilizarlo? ¿Puedes utilizar su resultado? ¿Puedes emplear su método? ¿Te hace falta introducir algún elemento auxiliar a fin de poder utilizarlo?

- ¿Puedes enunciar al problema de otra forma? ¿Puedes plantearlo en forma diferente nuevamente? Recurre a las definiciones.
- Si no puedes resolver el problema propuesto, trata de resolver primero algún problema similar. ¿Puedes imaginarte un problema análogo un
 tanto más accesible? ¿Un problema más general? ¿Un problema más
 particular? ¿Un problema análogo? ¿Puede resolver una parte del problema? Considera sólo una parte de la condición; descarta la otra parte;
 ¿en qué medida la incógnita queda ahora determinada? ¿En qué forma
 puede variar? ¿Puedes deducir algún elemento útil de los datos? ¿Puedes
 pensar en algunos otros datos apropiados para determinar la incógnita?
 ¿Puedes cambiar la incógnita? ¿Puedes cambiar la incógnita o los datos,
 o ambos si es necesario, de tal forma que estén más cercanos entre sí?
- ¿Has empleado todos los datos? ¿Has empleado toda la condición? ¿Has considerado todas las nociones esenciales concernientes al problema?

Paso 3: Ejecutar el plan

- Al ejercutar tu plan de la solución, comprueba cada uno de los pasos
- ¿Puedes ver claramente que el paso es correcto? ¿Puedes demostrarlo?

Paso 4: Examinar la solución obtenida

- ¿Puedes verificar el resultado? ¿Puedes el razonamiento?
- ¿Puedes obtener el resultado en forma diferente? ¿Puedes verlo de golpe? ¿Puedes emplear el resultado o el método en algún otro problema?
- G. Polya "Cómo plantear y resolver problemas" (Ed. Trillas, 1978) p. 19

B.2. Método de Pólya para resolver problemas de programación

Para resolver un problema se necesita:

Paso 1: Entender el problema

- ¿Cuáles son las argumentos? ¿Cuál es el resultado? ¿Cuál es nombre de la función? ¿Cuál es su tipo?
- ¿Cuál es la especificación del problema? ¿Puede satisfacerse la especificación? ¿Es insuficiente? ¿Redundante? ¿Contradictoria? ¿Qué restricciones se suponen sobre los argumentos y el resultado?
- ¿Puedes descomponer el problema en partes? Puede ser útil dibujar diagramas con ejemplos de argumentos y resultados.

Paso 2: Diseñar el programa

- ¿Te has encontrado con un problema semejante? ¿O has visto el mismo problema planteado en forma ligeramente diferente?
- ¿Conoces algún problema relacionado con éste? ¿Conoces alguna función que te pueda ser útil? Mira atentamente el tipo y trata de recordar un problema que sea familiar y que tenga el mismo tipo o un tipo similar.
- ¿Conoces algún problema familiar con una especificación similar?
- He aquí un problema relacionado al tuyo y que ya has resuelto. ¿Puedes utilizarlo? ¿Puedes utilizar su resultado? ¿Puedes emplear su método? ¿Te hace falta introducir alguna función auxiliar a fin de poder utilizarlo?
- Si no puedes resolver el problema propuesto, trata de resolver primero algún problema similar. ¿Puedes imaginarte un problema análogo un tanto más accesible? ¿Un problema más general? ¿Un problema más particular? ¿Un problema análogo?
- ¿Puede resolver una parte del problema? ¿Puedes deducir algún elemento útil de los datos? ¿Puedes pensar en algunos otros datos apropiados para determinar la incógnita? ¿Puedes cambiar la incógnita? ¿Puedes cambiar la incógnita o los datos, o ambos si es necesario, de tal forma que estén más cercanos entre sí?
- ¿Has empleado todos los datos? ¿Has empleado todas las restricciones sobre los datos? ¿Has considerado todas los requisitos de la especificación?

Paso 3: Escribir el programa

- Al escribir el programa, comprueba cada uno de los pasos y funciones auxiliares.
- ¿Puedes ver claramente que cada paso o función auxiliar es correcta?
- Puedes escribir el programa en etapas. Piensas en los diferentes casos en los que se divide el problema; en particular, piensas en los diferentes casos para los datos. Puedes pensar en el cálculo de los casos independientemente y unirlos para obtener el resultado final
- Puedes pensar en la solución del problema descomponiéndolo en problemas con datos más simples y uniendo las soluciones parciales para obtener la solución del problema; esto es, por recursión.
- En su diseño se puede usar problemas más generales o más particulares. Escribe las soluciones de estos problemas; ellas puede servir como guía para la solución del problema original, o se pueden usar en su solución.
- ¿Puedes apoyarte en otros problemas que has resuelto? ¿Pueden usarse? ¿Pueden modificarse? ¿Pueden guiar la solución del problema original?

Paso 4: Examinar la solución obtenida

- ¿Puedes comprobar el funcionamiento del programa sobre una colección de argumentos?
- ¿Puedes comprobar propiedades del programa?
- ¿Puedes escribir el programa en una forma diferente?
- ¿Puedes emplear el programa o el método en algún otro programa?

Simon Thompson *How to program it*, basado en G. Polya *Cómo plantear y resolver problemas*.

Bibliografía

- [1] J. A. Alonso and M. J. Hidalgo. Piensa en Haskell (Ejercicios de programación funcional con Haskell). Technical report, Univ. de Sevilla, 2012.
- [2] R. Bird. *Introducción a la programación funcional con Haskell*. Prentice–Hall, 1999.
- [3] H. C. Cunningham. Notes on functional programming with Haskell. Technical report, University of Mississippi, 2010.
- [4] H. Daumé. Yet another Haskell tutorial. Technical report, University of Utah, 2006.
- [5] A. Davie. *An introduction to functional programming systems using Haskell*. Cambridge University Press, 1992.
- [6] K. Doets and J. van Eijck. *The Haskell road to logic, maths and programming*. King's College Publications, 2004.
- [7] J. Fokker. Programación funcional. Technical report, Universidad de Utrech, 1996.
- [8] P. Hudak. *The Haskell school of expression: Learning functional programming through multimedia*. Cambridge University Press, 2000.
- [9] P. Hudak. The Haskell school of music (From signals to symphonies). Technical report, Yale University, 2012.
- [10] G. Hutton. *Programming in Haskell*. Cambridge University Press, 2007.
- [11] B. O'Sullivan, D. Stewart, and J. Goerzen. *Real world Haskell*. O'Reilly, 2008.
- [12] G. Pólya. Cómo plantear y resolver problemas. Editorial Trillas, 1965.
- [13] F. Rabhi and G. Lapalme. *Algorithms: A functional programming approach*. Addison-Wesley, 1999.

46 Bibliografía

[14] B. C. Ruiz, F. Gutiérrez, P. Guerrero, and J. Gallardo. *Razonando con Haskell (Un curso sobre programación funcional)*. Thompson, 2004.

[15] S. Thompson. *Haskell: The craft of functional programming*. Addison-Wesley, third edition, 2011.